Aclosure[RATS]

Johan G. F. Belinfante
2012 July 29

In[1]:= **SetDirectory["l:"]; << goedel.12jul28a**

```
:Package Title: goedel.12jul28a                 2012 July 28 at 3:50 p.m.

Loading takes about sixteen minutes, half that time due to builtin pauses.

It is now:  2012 Jul 29 at 5:53

Loading Simplification Rules

TOOLS.M is now incorporated in the GOEDEL program as of 2010 September 3

weightlimit = 40

Loading completed.

It is now:  2012 Jul 29 at 6:8
```

summary

The set of all intersections of sets of rational numbers is the union of the set of rational numbers and the singleton of the origin of the plane **Z** × **Z**.

a lower bound for Aclosure[RATS]

The intersection **A[x]** of all members of a class **x** is a set unless **x** is empty. The class of all **A[y]** for all non-empty subsets **y** of a class **x** is **Aclosure[x]**. If **x** is a set then **Aclosure[x] = fix[HULL[x]]**. Here **HULL[x]** is the function that takes each set **y ∈ image[inverse[S], x]** to the intersection of all members **w ∈ x** such that **y ⊂ w**. A lower bound for **Aclosure[RATS]** can be derived using the fact that **A[RATS] = {id[ω]} × {id[ω]}**.

In[9]:= **A[RATS]**

Out[9]= cart[set[id[omega]], set[id[omega]]]

Theorem.

In[7]:= **SubstTest[subclass, set[A[t]], fix[HULL[t]], t → RATS] // Reverse**

Out[7]= equal[cart[set[id[omega]], set[id[omega]]],
  hull[RATS, cart[set[id[omega]], set[id[omega]]]]] == True

In[8]:= **hull[RATS, cart[set[id[omega]], set[id[omega]]]] := cart[set[id[omega]], set[id[omega]]]**

Observation. The above result implies that the set **RATS ⋃ {A[RATS]}** is a lower bound for **Aclosure[RATS]**.

```
In[10]:= subclass[union[RATS, set[A[RATS]]], Aclosure[RATS]]

Out[10]= True
```

It will be shown below that this inclusion is actually an equation. What one needs to show is that the intersection of any set of rationals with two or more members is the origin of the plane **Z** × **Z**.

Theorem. The origin belongs to the intersection of any set of rationals. (For the empty set, the intersection is **V**.)

```
In[11]:= SubstTest[implies, subclass[x, y], subclass[A[y], A[x]], y → RATS] // Reverse

Out[11]= or[member[pair[id[omega], id[omega]], A[x]], not[subclass[x, RATS]]] == True

In[12]:= or[member[pair[id[omega], id[omega]], A[x_]], not[subclass[x_, RATS]]] := True
```

---

intersections of two rationals

Any two distinct rational numbers intersection only at the origin of **Z** × **Z**.

```
In[13]:= implies[and[member[x, RATS], member[y, RATS]],
          or[equal[x, y], equal[intersection[x, y], A[RATS]]]]

Out[13]= True
```

In this section a variable-free reformulation of this fact is derived.

Lemma. (Eliminating both variables.)

```
In[14]:= Map[empty[domain[complement[#]]] &, SubstTest[class, pair[x, y],
           implies[member[pair[x, y], u], member[pair[x, y], v]], {u → restrict[Di, RATS, RATS],
            v → image[inverse[CAP], set[cart[set[id[omega]], set[id[omega]]]]]}]]

Out[14]= subclass[cart[RATS, RATS],
           union[Id, image[inverse[CAP], set[cart[set[id[omega]], set[id[omega]]]]]]] == True

In[15]:= % /. Equal → SetDelayed
```

Theorem. A slightlty better variable-free reformulation.

```
In[16]:= SubstTest[implies, subclass[u, v], subclass[image[t, u], image[t, v]],
           {t → composite[CAP, id[Di]], u -> cart[RATS, RATS], v → union[Id,
              image[inverse[CAP], set[cart[set[id[omega]], set[id[omega]]]]]]}] // Reverse

Out[16]= subclass[image[CAP, composite[id[RATS], Di, id[RATS]]],
           set[cart[set[id[omega]], set[id[omega]]]]] == True

In[17]:= % /. Equal → SetDelayed
```

An equation would be even better.

Lemma. The reverse inclusion.

```
In[18]:= SubstTest[implies, subclass[u, v], subclass[image[t, u], image[t, v]],
          {t → CAP, u → set[PAIR[id[Z], cart[Z, set[id[omega]]]]],
           v -> composite[id[RATS], Di, id[RATS]]}] // Reverse

Out[18]= member[cart[set[id[omega]], set[id[omega]]],
          image[CAP, composite[id[RATS], Di, id[RATS]]]] == True

In[19]:= % /. Equal → SetDelayed
```

Theorem. A variable-free equation.

```
In[20]:= SubstTest[and, subclass[u, v], subclass[v, u],
         {u -> image[CAP, composite[id[RATS], Di, id[RATS]]],
          v -> set[cart[set[id[omega]], set[id[omega]]]]}]

Out[20]= equal[image[CAP, composite[id[RATS], Di, id[RATS]]],
          set[cart[set[id[omega]], set[id[omega]]]]] == True

In[21]:= image[CAP, composite[id[RATS], Di, id[RATS]]] :=
         set[cart[set[id[omega]], set[id[omega]]]]
```

intersections of 2 or more rationals

To complete the computation of **Aclosure[RATS]** one also needs to consider subsets of **RATS** with more than two elements. It will be shown that the intersection of any such set is the origin. This is because each such set **x** has a subset **y** with exactly two elements. The derivation to be presented below first introduces these two new variables **x** and **y**, and then eliminates them simultaneously using **reify** and **case**. It is actually convenient to assume just that **x** has at least two elements, rather than that **x** has more than two elements. The following theorem is a result of the observation that the class of two-element sets is a subset of the class of sets with at least two elements.

Theorem. Intersections of two or more rationals.

```
In[22]:= SubstTest[implies, subclass[u, v], subclass[image[t, u], image[t, v]],
          {t → BIGCAP, u → intersection[image[PAIRSET, Di], P[RATS]],
           v → intersection[image[S, image[PAIRSET, Di]], P[RATS]]}] // Reverse

Out[22]= member[cart[set[id[omega]], set[id[omega]]],
          image[BIGCAP, intersection[complement[range[SINGLETON]], P[RATS]]]] == True

In[23]:= % /. Equal → SetDelayed
```

The following statement introduces a variable into the variable-free result derived in the preceding section.

Theorem. The intersection of set of two different rationals is the singleton of the origin of **Z** × **Z**.

```
In[25]:= SubstTest[implies, and[member[x, y], not[member[0, y]]], member[A[x], image[BIGCAP, y]],
           y → intersection[P[RATS], image[PAIRSET, Di]]] // Reverse // MapNotNot

Out[25]= or[equal[A[x], cart[set[id[omega]], set[id[omega]]]],
           not[member[x, image[PAIRSET, Di]]], not[subclass[x, RATS]]] == True

In[26]:= or[equal[A[x_], cart[set[id[omega]], set[id[omega]]]],
           not[member[x_, image[PAIRSET, Di]]], not[subclass[x_, RATS]]] := True
```

Theorem. A special case of the general result than two classes are equal if each is a subclass of the other. Here one of the sets is the singleton of an ordered pair.

```
In[27]:= or[equal[z, cart[set[x], set[y]]], not[member[pair[x, y], z]],
           not[subclass[z, cart[set[x], set[y]]]]] // AssertTest

Out[27]= or[equal[z, cart[set[x], set[y]]],
           not[member[pair[x, y], z]], not[subclass[z, cart[set[x], set[y]]]]] == True

In[28]:= or[equal[z_, cart[set[x_], set[y_]]], not[member[pair[x_, y_], z_]],
           not[subclass[z_, cart[set[x_], set[y_]]]]] := True
```

Lemma. If **x** is a set of rationals containing a set **y** with two distinct members, then **A[x] = {id[ω]} × {id[ω]}**.

```
In[29]:= Map[not, SubstTest[and, implies[and[p3, p4], p5], implies[and[p2, p5], p6],
           implies[p1, p7], (*implies[and[p6,p7],p8],*) not[implies[and[p1, p2, p3, p4], p8]],
           {p1 → subclass[x, RATS], p2 → subclass[y, x], p3 → member[y, image[PAIRSET, Di]],
            p4 → subclass[y, RATS], p5 → equal[A[y], cart[set[id[omega]], set[id[omega]]]],
            p6 → subclass[A[x], cart[set[id[omega]], set[id[omega]]]],
            p7 → contains[A[x], cart[set[id[omega]], set[id[omega]]]],
            p8 → equal[A[x], cart[set[id[omega]], set[id[omega]]]]}]] // Reverse

Out[29]= or[equal[A[x], cart[set[id[omega]], set[id[omega]]]],
           not[member[y, image[PAIRSET, Di]]],
           not[subclass[x, RATS]], not[subclass[y, x]]] == True

In[30]:= (% /. {x → x_, y → y_}) /. Equal → SetDelayed
```

Theorem. (Eliminating both of the variables using **reify** and **case**.)

```
In[31]:= Map[equal[V, domain[#]] &,
          SubstTest[reify, x, case[or[equal[A[first[x]], o], not[member[second[x], p]],
            not[subclass[first[x], r]], not[subclass[second[x], first[x]]]]],
           {o -> cart[set[id[omega]], set[id[omega]]], p -> image[PAIRSET, Di], r → RATS}]]

Out[31]= subclass[image[BIGCAP, intersection[complement[range[SINGLETON]], P[RATS]]],
           set[cart[set[id[omega]], set[id[omega]]]]] == True

In[32]:= % /. Equal → SetDelayed
```

Corollary. An equation.

```
In[33]:= SubstTest[and, subclass[u, v], subclass[v, u],
          {u -> image[BIGCAP, intersection[complement[range[SINGLETON]], P[RATS]]],
           v -> set[cart[set[id[omega]], set[id[omega]]]]}]

Out[33]= equal[image[BIGCAP, intersection[complement[range[SINGLETON]], P[RATS]]],
           set[cart[set[id[omega]], set[id[omega]]]]] == True

In[34]:= image[BIGCAP, intersection[complement[range[SINGLETON]], P[RATS]]] :=
          set[cart[set[id[omega]], set[id[omega]]]]
```

Theorem. A formula for the class of all intersections of (nonempty) sets of rational numbers.

```
In[35]:= SubstTest[image, t, union[u, v], {t → BIGCAP, u → image[SINGLETON, RATS],
            v -> intersection[complement[range[SINGLETON]], P[RATS]]}] // Reverse

Out[35]= Aclosure[RATS] == union[RATS, set[cart[set[id[omega]], set[id[omega]]]]]

In[36]:= Aclosure[RATS] := union[RATS, set[cart[set[id[omega]], set[id[omega]]]]]
```